


Effect of Self Weight on a Cantilever Beam

Introduction

This tutorial was completed using ANSYS 7.0. The purpose of the tutorial is to show the required steps to account for the weight of an object in ANSYS.

Loads will not be applied to the beam shown below in order to observe the deflection caused by the weight of the beam itself. The beam is to be made of steel with a modulus of elasticity of 200 GPa.



Preprocessing: Defining the Problem

1. Give example a Title

Utility Menu > File > Change Title ...

/title, Effects of Self Weight for a Cantilever Beam

2. Open preprocessor menu

ANSYS Main Menu > Preprocessor

/PREP7

3. Define Keypoints

Preprocessor > Modeling > Create > Keypoints > In Active CS...

K, #, x, y, z

We are going to define 2 keypoints for this beam as given in the following table:

Keypoint	Coordinates (x,y,z)
1	(0,0)
2	(1000,0)

4. Create Lines

Preprocessor > Modeling > Create > Lines > Lines > In Active Coord

L, 1, 2

Create a line joining Keypoints 1 and 2

5. Define the Type of Element

Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the BEAM3 (Beam 2D elastic) element. This element has 3 degrees of freedom (translation along the X and Y axes, and rotation about the Z axis).

6. Define Real Constants

Preprocessor > Real Constants... > Add...

In the 'Real Constants for BEAM3' window, enter the following geometric properties:

- Cross-sectional area AREA: 500
- Area moment of inertia IZZ: 4166.67
- Total beam height: 10

This defines a beam with a height of 10 mm and a width of 50 mm.

7. Define Element Material Properties

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic

In the window that appears, enter the following geometric properties for steel:


- i. Young's modulus EX: 200000
 - ii. Poisson's Ratio PRXY: 0.3
8. **Define Element Density**
Preprocessor > Material Props > Material Models > Structural > Linear > Density
In the window that appears, enter the following density for steel:
- i. Density DENS: 7.86e-6
9. **Define Mesh Size**
Preprocessor > Meshing > Size Cntrl > ManualSize > Lines > All Lines...
For this example we will use an element edge length of 100mm.
10. **Mesh the frame**
Preprocessor > Meshing > Mesh > Lines > click 'Pick All'

Solution Phase: Assigning Loads and Solving

1. **Define Analysis Type**
Solution > Analysis Type > New Analysis > Static
ANTYPE, 0
2. **Apply Constraints**
Solution > Define Loads > Apply > Structural > Displacement > On Keypoints
Fix keypoint 1 (ie all DOF constrained)
3. **Define Gravity**

It is necessary to define the direction and magnitude of gravity for this problem.

- o Select **Solution > Define Loads > Apply > Structural > Inertia > Gravity...**
- o The following window will appear. Fill it in as shown to define an acceleration of 9.81m/s^2 in the y direction.




Note: Acceleration is defined in terms of meters (not 'mm' as used throughout the problem). This is because the units of acceleration and mass must be consistent to give the product of force units (Newtons in this case). Also note that a positive acceleration in the y direction stimulates gravity in the negative Y direction.

There should now be a red arrow pointing in the positive y direction. This indicates that an acceleration has been defined in the y direction.

```
DK, 1, ALL, 0,
ACEL, , 9.8
```

The applied loads and constraints should now appear as shown in the figure below.



4. Solve the System

Solution > Solve > Current LS
SOLVE


Postprocessing: Viewing the Results

1. Hand Calculations

Hand calculations were performed to verify the solution found using ANSYS:
The maximum deflection was shown to be 5.777mm

2. Show the deformation of the beam

General Postproc > Plot Results > Deformed Shape ... > Def + undef edge
PLDISP, 2



As observed in the upper left hand corner, the maximum displacement was found to be 5.777mm. This is in agreement with the theoretical value.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```

/Title, Effects of Self Weight
/PREP7

Length = 1000
Width = 50
Height = 10

K,1,0,0          ! Create Keypoints
K,2,Length,0

L,1,2

ET,1,BEAM3      ! Set element type
R,1,Width*Height,Width*(Height**3)/12,Height    !** = exponent
MP,EX,1,200000  ! Young's Modulus
MP,PRXY,1,0.3   ! Poisson's ratio
MP,DENS,1,7.86e-6 ! Density

LESIZE,ALL,Length/10, ! Size of line elements
LMESH,1             ! Mesh line 1

FINISH
/SOLU              ! Enter solution mode

ANTYPE,0          ! Static analysis
DK,1,ALL,0,       ! Constrain keypoint 1
ACEL,,9.8         ! Set gravity constant

SOLVE
FINISH

/POST1
PLDISP,2          ! Display deformed shape

```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>